Numerical and Symbolic Inverse Laplace Transform

Maxima Laplace

Initial value problem

$$[\omega_1\ \omega_2] := [-2\ 2]$$

$$ode := \begin{bmatrix} \frac{d^2}{dt^2} x(t) = \omega_1 \cdot x(t) + \omega_2 \cdot \frac{d}{dt} y(t) \\ \frac{d^2}{dt^2} y(t) = \omega_1 \cdot y(t) - \omega_2 \cdot \frac{d}{dt} x(t) \end{bmatrix}$$

$$ic := \begin{bmatrix} -1 & 1 \\ 1 & -1 \end{bmatrix} \qquad \begin{bmatrix} x(0) & x'(0) \\ y(0) & y'(0) \end{bmatrix}$$

$$fnc := \begin{bmatrix} x & X \\ y & Y \end{bmatrix} \qquad [\ d\text{-}domain\ \ s\text{-}domain\ ]$$

Symbolic Solver call $ODE := LapOde(ode, ic, fnc, t, s)$

Laplace transform of the ode

$$ODE_1 = \begin{bmatrix} -1 + s \cdot (1 + s \cdot X(s)) = -2 \cdot (1 - s \cdot Y(s) + X(s)) \\ 1 + s \cdot (-1 + s \cdot Y(s)) = -2 \cdot (1 + s \cdot X(s) + Y(s)) \end{bmatrix}$$

Freq-domain solution, used in numerical solution

$$ODE_2 = \begin{cases} X(s) = -\frac{(-1+s) \cdot s^2 + 2 \cdot (1 + 4 \cdot s)}{s^4 + 4 \cdot (1 + 2 \cdot s^2)} \\ Y(s) = \frac{(-1+s) \cdot s^2 + 2 \cdot (-3 + 2 \cdot s)}{s^4 + 4 \cdot (1 + 2 \cdot s^2)} \end{cases}$$

Time-domain solution: Maxima can't find a symbolic solution. Nor Wolfram alpha:

https://www.wolframalpha.com/input/?i=solve+x%27%27%3D-2*x%2B2*y%27+,+y%27%27%3D-2*y-2*x%27,+x(0

Plots: numerical solutions $[ti\ te\ N] := [0\ 4\ 20]$

$$\begin{bmatrix} Assign(ODE_2) \\ xo := ILap(X(s), ti, te, N) \\ x(t) := cinterp(xo, t) \\ yo := ILap(Y(s), ti, te, N) \\ y(t) := cinterp(yo, t) \end{bmatrix}$$



Using RK solver

$$ode = \begin{bmatrix} x'' = \omega_1 \cdot x + \omega_2 \cdot y' \\ y'' = \omega_1 \cdot y - \omega_2 \cdot x' \end{bmatrix} \qquad U = \begin{bmatrix} x \\ y \\ x' \\ y' \end{bmatrix} \qquad \frac{dU}{dt} = \begin{bmatrix} U_3 \\ U_4 \\ x'' \\ y'' \end{bmatrix}$$

$$D(t, U) := \begin{bmatrix} U_3 \\ U_4 \\ \omega_1 \cdot U_1 + \omega_2 \cdot U_4 \\ \omega_1 \cdot U_2 - \omega_2 \cdot U_3 \end{bmatrix}$$

Plot: compare numerical laplace invert (points) with RK solver (lines)

```
IC := stack(col(ic, 1), col(ic, 2))
xy := Rkadapt(IC, ti, te, 100, D(t, U))
tt := col(xy, 1)
xx := augment(tt, col(xy, 2))
x(t) := cinterp(xx, t)
yy := augment(tt, col(xy, 3))
y(t) := cinterp(yy, t)
```



Alvaro